
	Meist merkt man erst wie wichtig die Daten sind, wenn diese weg sind. Kauft man sich einen PC, wird zumeist kein Geld für Datensicherung investiert, wenn die Festplatte sich dann mit einem lauten Knattern verabschiedet, schaut man in die Röhre. 

In diesem Artikel finden Sie ein paar Ansätze, eine Datensicherung auf preiswerte Wechselmedien (z.B. ZIP, USB-Stift) oder CD/DVD-Brenner zu realisieren. Statt Wechselmedien könnte man auch einen über Netzwerk ansprechbares Laufwerk eines anderen Rechner verwenden. 

Eine Datensicherung auf die gleiche Festplatte im selben Rechner ist nonsens! "Stirbt" diese, sind sowohl die ursprünglichen, als auch die gesicherten Daten weg!

Eine Datensicherung auf eine andere Festplatte im selben Rechner ist nicht besonders glücklich: Liegt ein Controllerfehler vor, sind unter Umständen die Daten auf beiden Platten weg. 

Sinnvoll ist eine Datensicherung eigentlich nur dann, wenn Sie regelmäßig auf Wechselmedien erfolgt; vorsichtige Zeitgenossen nehmen die Medien aus dem Büro mit nach Hause, falls über Nacht das Büro abbrennt.

Alle "Datensicherung für Arme"-Ideen bestehen aus reinen Batch-Dateien mit einigen Zusatzprogrammen. Im folgenden sind die einzelnen Teile aufgeführt.



Dateien komprimieren
Eine oder mehrere Datei kopiert man mit dem Tool pkzip25:

Mit 

pkzip25 -add -recurse -path siwin.zip c:\windows\*.* 

würden Sie z.B. eine Archiv siwin erstellen, dass den kompletten Windows-Ordner mit allen Unterordnern enthält .

Mit 

pkzip25 -add -recurse -path -pass=hallo siwin.zip c:\windows\*.* 

würden Sie das gleiche Archiv siwin erstellen, die Dateien sind aber zusätzlich durch das Passwort "hallo" vor dem entpacken geschützt.

Mit 

pkzip25 -add -recurse -path sic.zip c:\dokume~1\admini~1\eigene~1\*.* 

sichert man z.B. die eigene Dateien des Administrators. 



Dateien überprüfen / löschen
Mit dem Kommando 

if exist dateiname  del dateiname
kann das Vorhandensein einer Datei überprüft werden und diese abhängig davon gelöscht werden.



Verzeichnisse löschen
Manchmal ist es sinnvoll, im Zielverzeichnis bestimmte Verzeichnisse zu löschen. Unter Win 9x gibt es hierzu den Befehlt deltree. Unter Windows NT/2000/XP, lässt sich das nur über rd realisieren.

Mit 

rd /s /q d:\lw_g\internet

löscht man das Verzeichnis d:\lw_g\internet mit sämtlichen Unterverzeichnissen unwiderruflich, das heißt die Löschung geht nicht über den Papierkorb.



Verzeichnisse kopieren
Verzeichnisse kopiert man am Besten mit XCOPY (und zwar in der DOS-Box, damit die langen Dateinamen erhalten bleiben).

xcopy g:\internet d:\lw_g\internet\ /s /e /c /r /h /k

Kopiert das Verzeichnis g:\internet nach d:\lw_g\internt und zwar mit allen Unterverzeichnissen. Die Parameter bewirken u.a., dass auch bei schreibgeschützen Dateien nicht einfach abgebrochen wird.



Wechselmedien überprüfen
Ab und zu kommt es vor, dass insbesondere ZIP-Dateien "spinnen". Mittels

chkdsk z: /f < ja.txt

kann man das Laufwerk Z: mittels CHKDSK überprüfen lassen (CHKDSK ist der Vorgänger von SCANDISK). Da CHKDSK eine Eingabe des Buchstabens "J" verlangt, muss man eine Datei im selben Verzeichnis anlegen, die einfach aus J und einem Zeilenumbruch besteht.

Unter Windows XP funktioniert 

chkdsk z: /f < ja.txt 
nicht - hier scheint 
chkdsk z: /f 
zu genügen.



Einen laufenden Prozess beenden
Oftmals ist es sinnvoll, einfach zur Sicherheit möglicherweise hängen gebliebene Backup-Prozesse zu beenden, damit ein Backup auch am Folgetag noch ausgeführt werden kann.

Hierfür kann man den Kill-Befehl verwenden, der z.B. im NT-Resource-Kit vorhanden ist oder den Sie hier herunterladen können. 

Mittels 

kill pkzip25
wird ein eventuell noch laufendes PKZIP25 beendet.



Registrierung sichern
Windows NT
Unter Windows NT kann man die Registrierung mittels

rdisk /s- 
sichern. Die Sicherungsdateien landen dabei um Ordner %systemroot%\repair. Von dort kann man sie mittels XCOPY bzw. PKZIP weiter sichern.

Windows 2000 / XP
rdisk gibt es nicht mehr. Man kann man sich mit erunt behelfen, welches unter http://home.t-online.de/home/lars.hederer/erunt/ zu finden ist.

Mittels 

erunt c:\backup /noconfirmdelete /nopropgresswindow
werden die Registry-Dateien nach C:\backup gesichert. Der Inhalt des Verzeichnisses wird dabei gelöscht! Ein Fortschrittsbalken wird nicht angezeigt.

Die Wiederherstellung der mit erunt gesicherten Registry ist unter http://home.t-online.de/home/lars.hederer/erunt/erunt-ger.txt beschrieben.

Win95
Bei jedem erfolgreichen Start werden Kopien der Registry-Dateien erstellt. Diese Dateien befinden sich im Windows-Verzeichnis, sind ebenfalls versteckt und heißen: SYSTEM.DA0 und USER.DA0.

Win98/ME 
Bei jedem ersten Start eines Tages werden Kopien der Registry-Dateien erstellt. Diese Dateien befinden sich im Verzeichnis Windows\Sysbckup und heißen Rb00n.cab. Voreingestellt sind fünf Sicherungen (n = 0 - 5). Die Anzahl der Sicherungen und die Dateien, die in der Sicherung enthalten sein sollen, können in der Datei Windows\Scanreg.ini festgelegt werden.
 

Diese Dateien können Sie in Ihrem Skript sichern. Eventuelle Schreibschutzattribute lassen sich mit 

attrib -r -h -s 
aufheben und mit

attrib +r +h +s
wieder setzen. 



Dienste anhalten - nur Windows NT/2000/XP
Wird z.B. ein Datenbankserver wie MS-SQL Server verwendet, müssen Sie diesen vor der Sicherung anhalten, da ansonsten die Daten nicht gesichert werden können. Dies passiert mit dem Befehl net stop

z.B. 

net stop mssqlserver
:
Sicherung ausführen

net start mssqlserver
net start sqlserveragent


Daten auf CD und DVD und brennen
Beim Brennprogramm Nero ist eine Version dabei, die es erlaubt, aus der Kommandozeile heraus einen Brennvorgang zu starten.

Mittels

c:\programme\ahead\nero\nerocmd --write --real --iso BackupCD --underrun_prot --close_session --import_iso_only --force_erase_cdrw --detect_non_empty_cdrw --nero_log_timestamp --drivename d c:\backup\backup.zip 
Wird die Datei c:\backup\backup.zip auf den CD-Brenner mit Laufwerksbuchstaben D: gebrannt.

Das ganze funktioniert auch mit DVD-Brennern

c:\programme\ahead\nero\nerocmd --create_udf_fs --dvd --write --real --iso BackupDVD --underrun_prot --close_session --import_iso_only --force_erase_cdrw --detect_non_empty_cdrw --nero_log_timestamp --recursive --drivename g C:\*.*
sichert das Laufwerk C: auf DVD-RAM (DVD-Brenner muss Laufwerk D sein, ansonsten ist --drivename zu ändern.

Nach der Sicherung geht die Schublade automatisch auf. Will man das verhindern, kann man 

--disable_eject 

verwenden.

Man kann dann z.B. mit 

dir z: /S > logdatei.txt

die Ausgabe als Logdatei für die Datensicherung benutzen.



Verwendung von Variablen
Auch Variable lassen sich in Batchdateien sinnvoll einsetzen. 

Beispiel:

set pzippr=c:\backup\pkzip25
set nero=c:\programme\ahead\nero\nerocmd
set log=c:\backup\backup.log
 

Bei langen Dateinamen mit Leerzeichen Anführungszeichen um die Variablen setzen!

set ziel=c:\backup\%date%.zip
Ziel wird auf einen Dateinamen gesetzt, der aus dem aktuellen Datum und der Endung ZIP besteht. 

Aufruf z.B. 

%pzippr% -add -recurse -path %ziel% C:\dokume~1\admini~1\eigene~1\*.* >> %log%
oder 

%nero% --write --real --iso BackupCD --underrun_prot --close_session --import_iso_only --force_erase_cdrw --detect_non_empty_cdrw --nero_log_timestamp --drivename d %ziel%
 



Konkrete Szenarien
Szenario 1
@echo off

REM Variablendeklaration

SET SOURCE=C:\dokume~1\admini~1\eigene~1\*.*

SET BACKUPPATH=C:\BACKUP\

REM

REM Letztes Backup löschen

cd %BACKUPPATH%

if exist 8.ZIP del 8.ZIP

REM VOrletztes Backup zum letzten machen ...

if exist 7.ZIP ren 7.ZIP 8.ZIP

if exist 6.ZIP ren 6.ZIP 7.ZIP

if exist 5.ZIP ren 5.ZIP 6.ZIP

if exist 4.ZIP ren 4.ZIP 5.ZIP

if exist 3.ZIP ren 3.ZIP 4.ZIP

if exist 2.ZIP ren 2.ZIP 3.ZIP

if exist 1.ZIP ren 1.ZIP 2.ZIP

REM Aktuelles Backup nach hinten schieben

REM Eigentliche Backup-Routine

%BACKUPPATH%pkZIP25 -add -recurse -path 1.ZIP %SOURCE%*.*

Source definiert das Verzeichnis, welches gesichert werden soll (die eigenen Dateien des Administrators). Backuppath ist der Zielpfad, auf dem die Sicherungsdateien landen sollen. 

Es wird zunächst geprüft, ob eine Datei namens 8.ZIP existiert, falls ja, wird diese gelöscht

Als zweites wird geprüft, ob eine Datei namens 7.ZIP exististiert, diese wird geg. in 8.ZIP umbenannt (REName). Als drittes wird geprüft, ob eine Datei namens 6.ZIP exististiert, diese wird geg. in 7.ZIP umbenannt (REName)....

Die eigentliche Sicherung erstellt eine Datei namens 1.ZIP. 

1.ZIP ist also immer die letzte, 2.ZIP die vor-letzte, 3.ZIP die vor-vor-letzte... Sicherung.

Szenario 2
@echo off

ECHO ************************************************************

ECHO ** DATENSICHERUNG MIT ANSCHLIESSENDER SICHERUNG AUF CD *****

ECHO ** CD EINLEGEN !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!! *****

ECHO ************************************************************

ECHO ************************************************************

ECHO ******* VORSICHT DATENBANK DIENST AM SERVER BEENDEN *******

ECHO ************************************************************

ECHO ************************************************************

pause

REM ****************************************************************

REM Backup-Programm auf CD-RW

REM Nach einer Idee aus c't 08/2003 09/2003 - umgesetzt von 

REM Lars Schlageter - lars-web.com

REM ****************************************************************

REM 

REM RANDBEDINGUNGEN:

REM - Im Verzeichnis C:\BACKUP muss das Programm pkzip25 installiert sein,

REM an sonsten ist die folgende Variabel anzupassen

set pzippr="c:\backup\pkzip25"

REM - AHEAD NERO muss als Brennprogramm verwendet werden

REM - NERO muss im Verzeichnis C:\PROGRAMME\AHEAD installiert sein, 

REM an sonsten ist die folgende Variable anzupassen:

set nero="c:\programme\ahead\nero\nerocmd"

REM Variable, die die Ziel-ZIP-Datei angibt

set ziel="c:\backup\%date%.zip"

REM Variable, die die LOG-Datei angibt

set log="c:\backup\backup.log"

REM Backup-Prozess

REM Schreibe Header in die LOG-Datei 

echo **************************************************************** > %log%

echo STARTE BACKUP: %date% %time% >> %log%

echo **************************************************************** >> %log%

echo **************************************************************** >> %log%

echo . >> %log%

echo . >> %log%

REM

REM Verzeichnis Nr 1

REM ****************

echo ********* >> %log%

echo D:\EIGENE DATEIEN >> %log%

%pzippr% -add -recurse -path %ziel% C:\dokume~1\admini~1\eigene~1\*.* >> %log%

REM Verzeichnis Nr 2

REM ****************

echo ********* >> %log%

echo C:\LEXWARE >> %log%

%pzippr% -add -recurse -path %ziel% c:\lexware\*.* >> %log%

REM

REM Datei brennen

%nero% --write --real --iso BackupCD --underrun_prot --close_session --import_iso_only --force_erase_cdrw --detect_non_empty_cdrw --nero_log_timestamp --drivename d %ziel%

cd c:\backup

echo . >> %log%

echo . >> %log%

echo . >> %log%

echo INHALT BACK

In diesem Szenario werden die Dateien aus C:\dokume~1\admini~1\eigene~1\*.* (Verzeichnis 1) und c:\lexware\ (Verzeichnis 2) ins Unterverzeichnis C:\BACKUP komprimiert. Die ZIP-Datei hat als Name das aktuelle Datum.

Anschließend wir die Datei auf Laufwerk D: gebrannt (D: muss natürlich ein Brenner sein). Die Kommandozeilenparameter sind so gesetzt, dass auch ein beschriebenes RW-Medium eingelegt werden kann, dieses wird automatisch von der Datensicherung gelöscht. 

Szenario 3
chkdsk z: /f < ja.txt
kill pkzip25
kill pkzip25
del z:\backup.zip
pkzip25 -add -recurse -path z:\backup.zip d:\winava\*.*
Diese Skript komprimiert den Inhalt von D:\WINAVA auf das Laufwerk Z in die Datei Z:\BACKUP. Laufwerk Z wird vorher mittels chkdsk überprüft.

Unter Windows XP funktioniert chkdsk z: /f < ja.txt nicht - hier scheint chkdsk z: /f zu genügen.

Szenario 4

@echo off

ECHO ************************************************************

ECHO ** DATENSICHERUNG MIT ANSCHLIESSENDER SICHERUNG AUF ZIP ****

ECHO ** ZIP-DISKETTE JETZT EINLEGEN !!!!!!!!!!!!!!!!!!!!!!!!!****

ECHO ************************************************************

ECHO LEXWARE UND ALLE ANDEREN PROGRAMME SCHLIESSEN!

ECHO GESICHERT WIRD NUR "LEXWARE" UND DER ORDNER EIGENE DATEIEN!

ECHO .

ECHO BITTE WARTEN SIE, BIS SICH DAS PROGRAMM BEENDET

set pzippr=c:\backup\pkzip25

set ziel=c:\backup\%date%.zip

set ziel2=c:\backup\%date%\

set ziel3=z:\%date%.zip

set log=c:\backup\backup.log

ECHO ..

PAUSE

REM

REM Backup-Prozess

REM Schreibe Header in die LOG-Datei 

echo **************************************************************** > %log%

echo STARTE BACKUP: %date% %time% >> %log%

echo **************************************************************** >> %log%

echo **************************************************************** >> %log%

echo . >> %log%

echo . >> %log%

ECHO ..BIN AM ARBEITEN

REM

REM SICHERE REGISTRY MITTELS ERUNT 

echo SICHERE REGISTRY >> %log%

c:\BACKUP\erunt %ziel2% /noconfirmdelete>> %log%

ECHO ..BIN AM ARBEITEN

REM Verzeichnis Nr 1

REM ****************

echo ********* >> %log%

echo D:\EIGENE DATEIEN >> %log%

%pzippr% -add -recurse -path %ziel% C:\dokume~1\user\eigene~1\*.* >> %log%

ECHO ..BIN AM ARBEITEN

REM

REM Verzeichnis Nr 2

REM ****************

echo ********* >> %log%

echo C:\LEXWARE >> %log%

%pzippr% -add -recurse -path %ziel% c:\lexware\*.* >> %log%

ECHO ..BIN AM ARBEITEN

REM

REM Verzeichnis Nr 3 - REGISTRY

REM ****************

echo ********* >> %log%

echo REGSITRY >> %log%

%pzippr% -add -recurse -path %ziel% %ziel2%*.* >> %log%

ECHO ..BIN AM ARBEITEN

REM

REM SICHERE DATEI AUF ZIP-DISKETTE

c:\windows\system32\chkdsk z: >> %log%

del z:\*.zip /Q >> %log%

copy %ziel% %ziel3% >> %log%

rd /s /q %ziel2% >> %log%

ECHO ......

ECHO FERTIG, BITTE šBERPRšFEN SIE DEN BERICHT UNTER %log%

PAUSE

Die Variablen:

pzippr = Das PKZIP25-Programm
ziel = Dateiname der Backup-Datei auf C:
ziel2 = Ordner in den die Registrierung gesichert wird.
ziel3 = Dateiname der Backup-Datei auf Z:
log= Dateiname der LOG-Datei

Zunächst wird mittels ERUNT die Registrierung in den Ordner %ziel2% gesichert.

Die Dateien aus dem Ordner Eigene Dateien (des User "user"), die Dateien aus dem Lexware und die Dateien aus dem Registrierungsordner %ziel2% werden in die Datei mit Namen %ziel% gepackt. Diese Datei wird zusätzlich auf Z: kopiert, wobei vorher alte ZIP-DATEIEN vom Z-Laufwerk gelöscht werden.

 

Szenario 4
@echo off

cls

c:\programme\ahead\nero\nerocmd --create_udf_fs --dvd --write --real --iso BackupCD --underrun_prot --close_session --import_iso_only --force_erase_cdrw --detect_non_empty_cdrw --nero_log_timestamp --recursive --drivename g C:\*.*

dir d: /s > "C:\Dokumente und Einstellungen\Administrator.SRV-2000MAX\Desktop\backup.log"

Sichert das Laufwerk C und erstellt die Ausgabe des Dir-Commandos als log-Datei.

Der Administrator kann die Gesamtsumme der gesicherten Dateien kontrollieren.



Suchwort DASI

